

Nyomógombok kezelése Arduino programokban

Nyomógomb használata:

- Adatbevitelhez
- Program futásának módosításához (menüvezérlés, kimenetek vezérlése, stb.)

Program működése:

- Nyomógomb lenyomásakor
- Nyomógomb felengedésekor
- Nyomógomb tartós lenyomva tartásakor
(időzítés, számláló pörgetés, stb.)
- Vegyes használat

Be- és kikapcsolási tranziensek kezelése

Gombnyomás által kiváltott feladat egyszeri végrehajtása

Nyomogomb_lenyomas.ino

A nyomógomb lenyomásakor hajtja végre egyszer a kijelölt feladatot:

```
unsigned long millis_nyomogomb = 0; //nyomógomb figyelése:20 msec
boolean gomb=0;
//*****

void setup(){
  pinMode(2, INPUT);    //nyomógomb
  pinMode(8, OUTPUT);    //LED gombnyomásakor
}
//*****

void loop(){
  nyomogomb();
}
//*****

void nyomogomb(){
  if (millis() - millis_nyomogomb > 20){
    millis_nyomogomb=millis();
    //program

    //Lenyomásakor kapcsol *****
    if (digitalRead(2)==HIGH) gomb=1;    //Ez a rész csak egyszer engedi a
    if (digitalRead(2)==LOW & gomb==1){ //következő utasításokat végrehajtani,
      gomb=0;                          //függetlenül a nyomva tartás idejétől.
      //program
      digitalWrite(8,HIGH);
      delay(10);
      digitalWrite(8,LOW);
    }
  }
}
//*****
```

Nyomogomb_felengedes.ino

A nyomógomb felengedésekor hajtja végre egyszer a kijelölt feladatot:

```
unsigned long millis_nyomogomb = 0; //nyomógomb figyelése:20 msec
boolean gomb=0;
//*****

void setup() {
  pinMode(2, INPUT);    //nyomógomb
  pinMode(9, OUTPUT);   //LED felengedéskor
}
//*****

void loop() {
  nyomogomb();
}
//*****

void nyomogomb(){
  if (millis() - millis_nyomogomb > 20){
    millis_nyomogomb=millis();
    //program

    //Felengedéskor kapcsol *****
    if (digitalRead(2)==LOW) gomb=1;
    if (digitalRead(2)==HIGH & gomb==1){
      gomb=0;
      //program
      digitalWrite(9,HIGH);
      delay(10);
      digitalWrite(9,LOW);
    }
  }
}
//*****
```

Nyomogomb_2sec_lenyomas.ino

A nyomógomb 2 másodpercig történő nyomva tartása után hajtja végre egyszer a kijelölt feladatot:

```
unsigned long millis_nyomogomb = 0; //nyomógomb figyelése:20 msec
unsigned long millis_lenyomva_tartas_2sec = 0;
boolean gomb=0;
boolean lenyomva_2sec=false;
boolean LED_allapot=LOW;
//*****
void setup() {
  pinMode(2, INPUT);    //nyomógomb
  pinMode(10, OUTPUT);  //LED 2 mp-ig lenyomva
}
//*****
void loop() {
  nyomogomb();
  lenyomva_tartas_2sec();
}
//*****
```

```

void nyomogomb(){
  if (millis() - millis_nyomogomb > 20){
    millis_nyomogomb=millis();
    //program

    //Lenyomásakor kapcsol *****
    if (digitalRead(2)==HIGH) gomb=1;    //Ez a rész csak egyszer engedi a
    if (digitalRead(2)==LOW & gomb==1){ //következő utasításokat végrehajtani,
      gomb=0;                          //függetlenül a nyomva tartás idejétől.
      //program

      millis_lenyomva_tartas_2sec=0;    //engedi a 2sec figyelést
    }
  }
}
//*****
void lenyomva_tartas_2sec(){
  if ((millis_lenyomva_tartas_2sec==0) && (digitalRead(2)==LOW)){
    millis_lenyomva_tartas_2sec=millis();
    lenyomva_2sec=true;
  }
  if (digitalRead(2)==HIGH){
    millis_lenyomva_tartas_2sec=0;
    lenyomva_2sec=false;
  }
  if ((millis() - millis_lenyomva_tartas_2sec > 2000) && (lenyomva_2sec==true)){
    lenyomva_2sec=false;
    //program
    LED_allapot = !LED_allapot;
    digitalWrite(10,LED_allapot);
  }
}
//*****

```

Nyomogomb_porgetes.ino

A nyomógomb megnyomásakor egyszer, majd 1 másodpercig történő nyomva tartása után 1/4 másodpercenként hajtja végre folyamatosan a kijelölt feladatot, a nyomógomb felengedéséig:

```
unsigned long millis_nyomogomb = 0; //nyomógomb figyelése:20 msec
unsigned long millis_lenyomva_tartas_1sec = 0;
unsigned long millis_porgetes_idozites=0;
boolean gomb=0;
boolean lenyomva_1sec=false;
//*****
void setup() {
  pinMode(3, INPUT);    //nyomógomb
  pinMode(11, OUTPUT);  //LED 1 mp-ig lenyomva, pörgetés
}
//*****
void loop() {
  nyomogomb();
  lenyomva_tartas_1sec(); //1 mp után pörgetés
}
//*****
```

```

void nyomogomb(){
  if (millis() - millis_nyomogomb > 20){
    millis_nyomogomb=millis();
    //program

    //Lenyomásakor kapcsol *****gomb*****
    if (digitalRead(3)==HIGH) gomb=1;    //Ez a rész csak egyszer engedi a
    if (digitalRead(3)==LOW & gomb==1){  //következő utasításokat végrehajtani,
      gomb=0;                          //függetlenül a nyomva tartás idejétől.
      //program
      digitalWrite(11,HIGH);
      delay(10);
      digitalWrite(11,LOW);
      millis_lenyomva_tartas_1sec=0;    //engedi az 1sec figyelést
    }
  }
}
//*****

```

```

void lenyomva_tartas_1sec(){
  if ((millis_lenyomva_tartas_1sec==0)&&(digitalRead(3)==LOW)){
    millis_lenyomva_tartas_1sec=millis();
    lenyomva_1sec=true;
  }
  if (digitalRead(3)==HIGH){
    millis_lenyomva_tartas_1sec=0;
    lenyomva_1sec=false;
  }
  if ((millis() - millis_lenyomva_tartas_1sec > 1000) && (lenyomva_1sec==true)){
    //program
    if ((millis() - millis_porgetes_idozites) >250){
      millis_porgetes_idozites=millis();
      //program
      digitalWrite(11,HIGH);
      delay(10);
      digitalWrite(11,LOW);
    }
  }
}
//*****

```

A **Nyomogomb_vegyes.ino** a fenti programokat együtt tartalmazza.

Nyomógombok csatlakoztatása: digitális 2-es és 3-as portra kell csatlakoztatni, a portok 10k ellenállással tápfeszültségre kötve. A nyomógombok másik csatlakozóját a testre kell kötni.

LED-ek csatlakoztatása: digitális 8, 9, 10 és 11-es portra a LED-ek anódját kell csatlakoztatni, 1k soros ellenállásokkal. A LED-ek katódját testre kell kötni.

